

无锡市北辰自动化技术有限公司

在 VB 中访问 NETLINK 并读写 S7300 PLC 数据
无锡市北辰自动化技术有限公司 张鹏 2007-10-16

一、概述

Hilscher NETLINK 用于将西门子 MPI 协议转换为以太网 TCP/IP 协议，并提供公开的无需授权的驱动函数库。

NETLINK 有三种应用：

1、对于仅仅需要 STEP7 编程监控和 WINCC 监控，安装 IBHNET127 驱动即可，该驱动安装后将在控制面板的 Set PG/PC Interface 中添加 IBHNet MPI/PPI/Profibus 三个驱动连接；将 STEP7 (S7 ONLINE) 和 WINCC (MPI) 的访问点指向以上驱动连接即可。

2、对于第三方组态软件（如 KingView、iFix 等）通过 NETLINK 访问 PLC，选择我公司开发的 OPCServer。

3、对于 VB 或者 VC++ (6.0 以上版本) 的自行开发界面，有两种方法实现通讯：

- 调用 Hilscher 提供的 IP Driver 驱动函数库进行通讯；
- 使用 winsock 控件进行通讯；

本文对在 VB6.0 中通过 Hilscher IP Driver 驱动库读写 S7300/400 PLC 数据的基本开发步骤作简要描述，具体资料可以参考 NETLINK 光盘 netDEVICE System Software 中的 Nlmpi_pie.pdf 和 Drv_Ip.pdf 文件，该光盘可到 http://www.gkong.com/co/beichen-automation/index_download.asp 页面中下载。


二、平台和配置

1、平台：WINDOW2000+SP4, Microsoft VB6.0;

2、软件和驱动安装：Hilscher SYCON 软件和 IP Driver;

3、硬件要求：NETLINK 电缆一根，西门子 S7300 CPU 一台（如 CPU314），以太网交换机一台（如 TP-LINK），以太网电缆；

4、系统配置：如下图所示



三、NETLINK 参数配置

如上图所示，将计算机网卡的 IP 地址配置为 192.168.1.10，将 NETLINK 的 RJ45 端口链接到交换机端口上，另一端插入 CPU314 的 MPI 通讯口，NETLINK 是由 CPU314 的 MPI 口供电的。如果 CPU 上已经连接了其他监控设备如触摸屏，则可以选择西门子带编程口的网络插头。

在 NETLINK 链接到 MPI 端口后的三分钟内（上电三分钟内）运行[开始]-[程序]-[Hilscher IP

Driver]-[NetIdent Demo Program]，点击[Start Poll]按钮自动查找网络上的 NETLINK，找到后将会显示 NETLINK 信息，NETLINK 出厂默认 IP 地址为 0.0.0.0，选择显示的 NETLINK 并点击[Set IP]按钮，设置 IP 地址为 192.168.1.88；如果软件找不到 NETLINK，请用 PING 命令检查网络是否链接正确。

注意 1：通过 NetIdent 设定的 IP 地址是临时的，在 NETLINK 再次上电后将自动恢复到原来的 IP 地址；

注意 2：NETLINK 上电的三分钟内允许通过 NetIdent 设定其 IP 地址，三分钟后设定将被拒绝；

设定好 NETLINK 的 IP 地址后运行[开始]-[程序]-[SyCon System Configurator]，新建项目并添加 NETLINK 设备，双击 NETLINK 并设定网络参数，IP 地址设为 192.168.1.88，选择自动检测接口和自动速率识别，NETLINK 的 MPI 地址设定为 1（注意不要和 MPI 网络上其他站点地址冲突）；设定好参数后将其下载到 NETLINK。（具体可以参考《NetLINK 和 IBHNet 驱动在 SIEMENS STEP7 WINCC 中的使用说明.doc》中的描述）

注意 3：通过 SYCON 软件设定的 IP 地址是永久固定的，在 NETLINK 再次上电后不会改变；

无锡市滴翠路100号国家工业设计园创意园A幢903室 邮编: 214072
电话: (0510) 85166813, 85166823 传真: (0510) 85166813 Email: wxbca@163.com

无锡市北辰自动化技术有限公司

四、IP Driver 参数配置

运行[开始]-[程序]-[Hilscher IP Driver]-[IP Driver Setup], 在 Connection 0 中输入 NETLINK 的 IP 地址: 192.168.1.88、1099、TCP、Client、5000;


运行[开始]-[程序]-[Hilscher IP Driver]-[IP Driver Test], 在菜单[Connection select]中选择 Connection 0, 连接后进行其他菜单的测试, 如[Driver Info]等, 如果测试正常说明 IP Driver 已连接到 NETLINK。

五、VB 应用

应用举例:

- 1、按下[循环启动]按钮后周期性读取 MWO 的数值(间隔 100ms), 直到按下[停止]按钮。
- 2、按下[写入新值]按钮命令将新值写入 MWO。
- 3、[复位 NETLINK]按钮初始化 NETLINK。

演示界面如下图所示:



开发步骤:

1、新建 VB EXE 工程, 将 NETLINK 光盘中 Api\Demo\VBASIC32\CIFUSER.BAS 模板文件复制到 VB 工程中并取消只读属性; 然后添加到工程中, 打开 CIFUSER.BAS 模板, 加入以下常数申明:

```
' 新增常数
Public Const TASK_TDT_UINT8 = 5
Public Const TASK_TFC_READ = 1
Public Const TASK_TFC_WRITE = 2
```

2、新建模块 Module1, 添加以下公共变量和 API 函数:

```
Public Const usSize = 288 ' 消息长度常数=288
Public SndMsgRead As MSG_STRUC ' 读取 MWO 数据的消息
Public SndMsgWrite As MSG_STRUC ' 写 MWO 数据的消息
Public SndMsgReset As MSG_STRUC ' 复位 NETLINK 的消息
Public RcvMsg As MSG_STRUC ' 接收消息
' 内存拷贝 API
Public Declare Sub CopyMemory Lib "kernel32" Alias "RtlMoveMemory" (Destination As Any,
Source As Any, ByVal Length As Long)
```

3、在 Form1 代码窗口中添加读写控制变量、三个消息初始化函数、NETLINK 链接和读写函数(请参考 Nlmpi_pie.pdf 文件):

```
Dim bNetLink_Connected As Boolean ' NETLINK 链接标记
Dim bStartRead As Boolean ' 开始循环读取标记
Dim bWrite As Boolean ' 写数据标记

' 初始化读 MWO 数据的消息
Private Function Initial_SndMsgRead()
SndMsgRead.rx = 3 ' 接收号=3
```

无锡市滴翠路100号国家工业设计园创意园A幢903室 邮编: 214072
电话: (0510) 85166813, 85166823 传真: (0510) 85166813 Email: wxbca@163.com

```

SndMsgRead.tx = 255      ' 发送号=255
SndMsgRead.ln = 8       ' 读数据长度=8
SndMsgRead.nr = 0       ' 消息号
SndMsgRead.a = 0        ' 应答号=0
SndMsgRead.f = 0        ' 错误号=0
SndMsgRead.b = 0        ' 命令码=16#33(M区数据)
SndMsgRead.e = 0        ' 扩展号=0
SndMsgRead.data(0) = 2  ' S7300 CPU 的 MPI 地址=2
SndMsgRead.data(1) = 0  ' 数据区=0
SndMsgRead.data(2) = 0  ' M区起始地址=0(起始地址 MOD 256)
SndMsgRead.data(3) = 0  ' M区起始地址=0(起始地址 / 256)
SndMsgRead.data(4) = 0  ' 数据索引=0(仅用于 DB 区)
SndMsgRead.data(5) = 2  ' 读取的字节数=2 (MW0)
SndMsgRead.data(6) = TASK_TDT_UINT8  ' 数据类型=TASK_TDT_UINT8(字节)
SndMsgRead.data(7) = TASK_TFC_READ   ' 功能号=TASK_TFC_READ(读数据)
End Function

```

```

' 初始化写 MW0 数据的消息
Private Function Initial_SndMsgWrite()
SndMsgWrite.rx = 3      ' 接收号
SndMsgWrite.tx = 255    ' 发送号
SndMsgWrite.ln = 10     ' 写数据长度=10
SndMsgWrite.nr = 0      ' 消息号
SndMsgWrite.a = 0       ' 应答号=0
SndMsgWrite.f = 0       ' 错误号=0
SndMsgWrite.b = 0       ' 命令码=16#33(M区数据)
SndMsgWrite.e = 0       ' 扩展号=0
SndMsgWrite.data(0) = 2 ' S7300 CPU 的 MPI 地址=2
SndMsgWrite.data(1) = 0 ' 数据区=0
SndMsgWrite.data(2) = 0 ' M区起始地址=0(起始地址 MOD 256)
SndMsgWrite.data(3) = 0 ' M区起始地址=0(起始地址 / 256)
SndMsgWrite.data(4) = 0 ' 数据索引=0(仅用于 DB 区)
SndMsgWrite.data(5) = 0 ' 要写的字节数=2 (MW0)
SndMsgWrite.data(6) = TASK_TDT_UINT8  ' TASK_TDT_UINT8
SndMsgWrite.data(7) = TASK_TFC_WRITE   ' TASK_TFC_WRITE
SndMsgWrite.data(8) = 0  ' 待写数据低字节=0
SndMsgWrite.data(9) = 0  ' 待写数据高字节=0
End Function

```

```

' 初始化复位 NETLINK 消息
Private Function Initial_SndMsgReset()
SndMsgReset.rx = 0      ' 接收号=0 (NETLINK 系统)
SndMsgReset.tx = 255    ' 发送号
SndMsgReset.ln = 1      ' 数据长度
SndMsgReset.nr = 1      ' 消息号
SndMsgReset.a = 0       ' 应答号=0
SndMsgReset.f = 0       ' 错误号=0
SndMsgReset.b = 1       ' 命令码=1
SndMsgReset.e = 0       ' 扩展号=0
SndMsgReset.data(0) = 1 ' =1
End Function

```

```

' NETLINK 链接
Private Function NetLINK_Connect ()
Dim sRet As Integer
sRet = DevOpenDriver(0) ' 打开驱动
If sRet = DRV_NO_ERROR Then
    sRet = DevInitBoard(0, 0) ' 初始化板卡
    If sRet = DRV_NO_ERROR Then
        bNetLink_Connected = True ' 设置 NETLINK 链接标记
    Else
        MsgBox "找不到 NETLINK!", vbCritical + vbOKOnly, "错误"
        DevExitBoard 0 ' 退出板卡
        DevCloseDriver 0 ' 关闭驱动
    End If
Else
    MsgBox "找不到 NETLINK 驱动!", vbCritical + vbOKOnly, "错误"
    DevExitBoard 0
    DevCloseDriver 0
End If
End Function

```

```

' 读 MWO 数据
Private Function ReadValue ()
Dim sRet As Integer, RcvData(0 To 1) As Byte, Val_INT16 As Integer
sRet = DevPutMessage(0, SndMsgRead, 500) ' 发送读取数据的消息
If sRet = DRV_NO_ERROR Then ' 如果发送消息无错误
    sRet = DevGetMessage(0, usSize, RcvMsg, 500) ' 接收返回消息
    ' 如果接收无错误且消息无错误且功能号为读取消息则接收消息中的 MWO 数据
    If sRet = DRV_NO_ERROR And RcvMsg.f = 0 And RcvMsg.data(7) = TASK_TFC_READ Then
        RcvData(0) = RcvMsg.data(9) ' 注意高低字节的颠到
        RcvData(1) = RcvMsg.data(8)
        CopyMemory Val_INT16, RcvData(0), 2
        Text_Value.Text = Val_INT16
    End If
End If
End Function

```

```

' 写 MWO 数据
Private Function WriteValue ()
Dim sRet As Integer, SndData(0 To 1) As Byte, Val_INT16 As Integer
Val_INT16 = Int(Text_Setting.Text)
CopyMemory SndData(0), Val_INT16, 2
SndMsgWrite.data(8) = SndData(1) ' 注意高低字节的颠到
SndMsgWrite.data(9) = SndData(0)
sRet = DevPutMessage(0, SndMsgWrite, 500) ' 发送写数据的消息
If sRet = DRV_NO_ERROR Then ' 如果发送消息无错误
    sRet = DevGetMessage(0, usSize, RcvMsg, 500) ' 接收返回消息
End If
bWrite = False ' 复位写值标志
End Function

```

4、在 Form1 添加定时器 Timer1，设定周期为 100ms；在 Timer1 的执行事件中添加以下代码：

```

' 定时器 1 的执行代码
Private Sub Timer1_Timer ()

```

```

If bNetLink_Connected = False Then ' 如果 NetLink 没有链接则退出定时器执行
    Exit Sub
End If
If bWrite = True Then ' 如果有写值请求
    WriteValue ' 写 MWO 值
ElseIf bStartRead = True Then ' 否则检查是否启动了读循环
    ReadValue ' 读 MWO 值
End If
End Sub

```

5、 添加[循环读取]、[停止]、[写入新值]、[复位 NETLINK]按钮的动作

```

' 循环读取按钮
Private Sub Command_read_Click()
If bNetLink_Connected = False Then ' 如果 NETLINK 未链接
    NetLINK_Connect ' 链接 NETLINK
End If
If bNetLink_Connected = True Then ' 如果已链接
    bStartRead = True ' 启动循环读
End If
End Sub

' 停止按钮
Private Sub Command_stop_Click()
bStartRead = False
End Sub

' 写入新值按钮
Private Sub Command_set_Click()
If IsNumeric(Text_Setting.Text) Then
    bWrite = True
Else
    MsgBox "设定值错误，范围：-32768 至 32767", vbExclamation + vbOKOnly, "格式错误"
End If
End Sub

' 复位 NETLINK 按钮
Private Sub Command_reset_Click()
Dim sRet As Integer
If bStartRead = True Then
    MsgBox "复位 NETLINK 前请先停止循环读取。", vbInformation + vbOKOnly, "提示"
Else
    sRet = DevPutMessage(0, SndMsgReset, 500) ' 发送复位 NETLINK 的消息
End If
End Sub

```

6、 在 Form1 的 Load 事件中添加初始化代码：

```

' Form1 装载
Private Sub Form_Load()
bStartRead = False
bWrite = False
bNetLink_Connected = False
Text_Value.Text = 0
Text_Setting.Text = 0
Initial_SndMsgRead ' 初始化读取数据消息

```

无锡市北辰自动化技术有限公司

```

Initial_SndMsgWrite      ' 初始化写值消息
Initial_SndMsgReset     ' 初始化复位消息
NetLINK_Connect        ' 链接 NETLINK
End Sub

```

六、注意事项

- 1、注意发送消息中的数据起始地址的高低字节颠倒和接收消息中数据区的高低字节颠倒；
- 2、定时器事件在操作系统繁忙时容易被中断执行，可以适当提高进程和线程的优先级来保证数据的更新速率，在 Form_Load() 中添加以下代码：

```

Dim hThread As Long, hProcess As Long
hThread = GetCurrentThread
hProcess = GetCurrentProcess
SetThreadPriority hThread, THREAD_PRIORITY_HIGHEST
SetPriorityClass hProcess, HIGH_PRIORITY_CLASS
在 Module1 中添加以下声明：

```

```

Public Const THREAD_BASE_PRIORITY_IDLE = -15
Public Const THREAD_BASE_PRIORITY_LOWRT = 15
Public Const THREAD_BASE_PRIORITY_MIN = -2
Public Const THREAD_BASE_PRIORITY_MAX = 2
Public Const THREAD_PRIORITY_LOWEST = THREAD_BASE_PRIORITY_MIN
Public Const THREAD_PRIORITY_HIGHEST = THREAD_BASE_PRIORITY_MAX
Public Const THREAD_PRIORITY_BELOW_NORMAL = (THREAD_PRIORITY_LOWEST + 1)
Public Const THREAD_PRIORITY_ABOVE_NORMAL = (THREAD_PRIORITY_HIGHEST - 1)
Public Const THREAD_PRIORITY_IDLE = THREAD_BASE_PRIORITY_IDLE
Public Const THREAD_PRIORITY_NORMAL = 0
Public Const THREAD_PRIORITY_TIME_CRITICAL = THREAD_BASE_PRIORITY_LOWRT
Public Const HIGH_PRIORITY_CLASS = &H80
Public Const IDLE_PRIORITY_CLASS = &H40
Public Const NORMAL_PRIORITY_CLASS = &H20
Public Const REALTIME_PRIORITY_CLASS = &H100

```

```

Public Declare Function SetThreadPriority Lib "kernel32" (ByVal hThread As Long, ByVal
nPriority As Long) As Long

```

```

Public Declare Function SetPriorityClass Lib "kernel32" (ByVal hProcess As Long, ByVal
dwPriorityClass As Long) As Long

```

- 3、无论是读取数据还是写数据，发送和接收消息函数必须成对出现；
- 4、在实际的通讯中如果网络中断（譬如交换机断电，插拔网络插头）将会导致发送和接收消息函数出错，其返回值 sRet 大于等于 10000（网络错误），此时需要主动发出 NETLINK 复位消息；发出复位消息后一般需要等待 5 秒钟后再次进行链接；
- 5、在每次尝试 NETLINK 链接前为了避免找不到 NETLINK 导致的驱动函数超时返回，可以先采用 NetIdent 协议（参考 Ni_pie.pdf 文件，使用 Winsock 控件）广播网络上的 NETLINK，找到 NETLINK 后再尝试链接，这样可以提高通讯效率；
- 6、一次读取地址连续的数据包（不要超过 240 个字节）比多次读取单个数据的效率要高的多；
- 7、发送和返回的响应消息的消息号 Nr 是对应的，可以用 Nr 来判断多个读写任务的响应消息；
- 8、如果一台计算机需要同时访问多个 NETLINK 通讯，建议采用 VC++6.0 开发，采用多线程技术；

七、联系我们

无锡市北辰自动化技术有限公司
 江苏省滴翠路 100 号无锡市国家工业设计园创意园 A 幢 903
 电话：0510-85166823, 85162669, 联系人：王海波，张鹏
 传真：0510-85166813
 Email: beichen@beichen-automation.com, 网址: www.beichen-automation.com

无锡市滴翠路100号国家工业设计园创意园A幢903室 邮编: 214072
 电话: (0510) 85166813, 85166823 传真: (0510) 85166813 Email: wxbca@163.com